New Solutions to the n-Body Problem via Nonlinear Programming

Robert J. Vanderbei (joint with Chetan Narain)

2009 October 11

INFORMS 2009

San Diego, CA

http://www.princeton.edu/∼rvdb

Celestial Mechanics

The N-Body Problem

Least Action Principle

Given: n bodies.

Let:

m_j denote the mass and
$z_j(t)$ denote the position in $\mathbb{R}^2 = \mathbb{C}$ of body j at time t.

Action Functional:

$$A = \int_0^{2\pi} \left(\sum_j \frac{m_j}{2} \|\dot{z}_j\|^2 + \sum_{j,k:k<j} \frac{m_j m_k}{\|z_j - z_k\|} \right) dt.$$

Critical points satisfy equations of motion.

Minimize!

Equations of Motion

First Variation:

$$\begin{aligned}
\delta A &= \int_0^{2\pi} \sum_\alpha \left(\sum_j m_j \dot{z}_j^\alpha \delta \dot{z}_j^\alpha - \sum_{j,k:k<j} m_j m_k \frac{(z_j^\alpha - z_k^\alpha)(\delta z_j^\alpha - \delta z_k^\alpha)}{\|z_j - z_k\|^3} \right) dt \\
&= -\int_0^{2\pi} \sum_j \sum_\alpha \left(m_j \ddot{z}_j^\alpha + \sum_{k:k\neq j} m_j m_k \frac{z_j^\alpha - z_k^\alpha}{\|z_j - z_k\|^3} \right) \delta z_j^\alpha dt
\end{aligned}$$

Setting first variation to zero, we get:

$$m_j \ddot{z}_j^\alpha = -\sum_{k:k\neq j} m_j m_k \frac{z_j^\alpha - z_k^\alpha}{\|z_j - z_k\|^3}, \qquad j = 1, 2, \ldots, n, \quad \alpha = 1, 2$$

Note: If $m_j = 0$ for some j, then the first order optimality condition reduces to $0 = 0$, which is *not* the equation of motion for a massless body.

Periodic Solutions

We assume solutions can be expressed in the form

$$z_j(t) = \sum_{k=-\infty}^{\infty} \gamma_k e^{ikt}, \qquad \gamma_k \in \mathbb{C}.$$

Writing with components $z_j(t) = (x_j(t), y_j(t))$ and $\gamma_k = (\alpha_k, \beta_k)$, we get

$$\begin{aligned} x(t) &= a_0 + \sum_{k=1}^{\infty} \left(a_k^c \cos(kt) + a_k^s \sin(kt) \right) \\ y(t) &= b_0 + \sum_{k=1}^{\infty} \left(b_k^c \cos(kt) + b_k^s \sin(kt) \right) \end{aligned}$$

where

$$\begin{aligned} a_0 = \alpha_0, \qquad a_k^c &= \alpha_k + \alpha_{-k}, \qquad a_k^s = \beta_{-k} - \beta_k, \\ b_0 = \beta_0, \qquad b_k^c &= \beta_k + \beta_{-k}, \qquad b_k^s = \alpha_k - \alpha_{-k}. \end{aligned}$$

The variables a_0, a_k^c, a_k^s, b_0, b_k^c, and b_k^s are the decision variables in the optimization model.

The AMPL Model. Too hard!!

```
param N := 3;  # number of masses

param m := 100;  # number of terms in numerical approx to integral




var x {i in 0..N-1, j in 0..m-1};

var y {i in 0..N-1, j in 0..m-1};


var xdot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (x[i,j+1]-x[i,j])*m/(2*pi) else (x[i,0]-x[i,m-1])*m/(2*pi);
var ydot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (y[i,j+1]-y[i,j])*m/(2*pi) else (y[i,0]-y[i,m-1])*m/(2*pi);

var Kinetic {j in 0..m-1} = 0.5*sum {i in 0..N-1} (xdot[i,j]^2 + ydot[i,j]^2);

var Potential {j in 0..m-1} = - sum {i in 0..N-1, ii in 0..N-1: ii>i}
                                  1/sqrt((x[i,j]-x[ii,j])^2 + (y[i,j]-y[ii,j])^2);

minimize Action: (2*pi/m)*sum {j in 0..m-1} (Kinetic[j] - Potential[j]);
```

The AMPL Model. Tractable

```
param N := 3;  # number of masses
param n := 15;  # number of terms in Fourier series representation
param m := 100;  # number of terms in numerical approx to integral

param theta {j in 0..m-1} := j*2*pi/m;

param a0 {i in 0..N-1} default 0;         param b0 {i in 0..N-1} default 0;
var as {i in 0..N-1, k in 1..n} := 0;     var bs {i in 0..N-1, k in 1..n} := 0;
var ac {i in 0..N-1, k in 1..n} := 0;     var bc {i in 0..N-1, k in 1..n} := 0;

var x {i in 0..N-1, j in 0..m-1}
  = a0[i]+sum {k in 1..n} ( as[i,k]*sin(k*theta[j]) + ac[i,k]*cos(k*theta[j]) );
var y {i in 0..N-1, j in 0..m-1}
  = b0[i]+sum {k in 1..n} ( bs[i,k]*sin(k*theta[j]) + bc[i,k]*cos(k*theta[j]) );

var xdot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (x[i,j+1]-x[i,j])*m/(2*pi) else (x[i,0]-x[i,m-1])*m/(2*pi);
var ydot {i in 0..N-1, j in 0..m-1}
  = if (j<m-1) then (y[i,j+1]-y[i,j])*m/(2*pi) else (y[i,0]-y[i,m-1])*m/(2*pi);

var Kinetic {j in 0..m-1} = 0.5*sum {i in 0..N-1} (xdot[i,j]^2 + ydot[i,j]^2);

var Potential {j in 0..m-1} = - sum {i in 0..N-1, ii in 0..N-1: ii>i}
                                   1/sqrt((x[i,j]-x[ii,j])^2 + (y[i,j]-y[ii,j])^2);

minimize Action: (2*pi/m)*sum {j in 0..m-1} (Kinetic[j] - Potential[j]);
```

Continued...

```
let {i in 0..N-1, k in 1..n} as[i,k] := 1*(Uniform01()-0.5);
let {i in 0..N-1, k in 1..n} ac[i,k] := 1*(Uniform01()-0.5);
let {i in 0..N-1, k in n..n} bs[i,k] := 0.01*(Uniform01()-0.5);
let {i in 0..N-1, k in n..n} bc[i,k] := 0.01*(Uniform01()-0.5);

solve;
```

Choreographies and the Ducati

The previous AMPL model was used to find many *choreographies* (a la Moore and Montgomery/Chencinier) in the equimass n-body problem and the stable *Ducati* solution to the 3-body problem.



http://www.princeton.edu/~rvdb/JAVA/astro/galaxy/Galaxy0.html

Ducatis with Unequal Masses





Floquet Stability Analysis

For simplicity, in this section we assume that all masses are equal to one. Let $\xi^*(t) = (z^*(t), \dot{z}^*(t))$ be a particular solution to

$$\dot{\xi} = A(\xi)$$

where

$$A\left(z(t), \dot{z}(t)\right) = (\dot{z}(t), a(z(t)))$$

and

$$a(z) = (a_1(z), \ldots, a_n(z))$$

and

$$a_j(z) = -\sum_{k:k\neq j} \frac{z_j - z_k}{\|z_j - z_k\|^2}, \quad j = 1, 2, \ldots, n.$$

Stability Analysis: Continued

Consider a nearby solution $\xi(t)$:

$$\begin{aligned}\dot{\xi}(t) &= A(\xi(t)) \\ &\approx A(\xi^*(t)) + A'(\xi^*(t))(\xi(t) - \xi^*(t)) \\ &= \dot{\xi}^*(t) + A'(\xi^*(t))(\xi(t) - \xi^*(t)).\end{aligned}$$

Put $\Delta\xi = \xi - \xi^*$. Then $\dot{\Delta\xi} = A'(\xi^*(t))\Delta\xi$. A finite difference approximation yields

$$\begin{aligned}\Delta\xi(t+h) &= \Delta\xi(t) + hA'(\xi^*(t))\Delta\xi(t) \\ &= \big(I + hA'(\xi^*(t))\big)\,\Delta\xi(t).\end{aligned}$$

Iterating around one period, we get:

$$\Delta\xi(T) = \left(\prod_{i=0}^{n-1}\big(I + hA'(\xi^*(t_i))\big)\right)\Delta\xi(0),$$

where $h = T/n$ and $t_i = iT/n$.
Orbit is *stable* if no eigenvalue lies outside the unit circle in the complex plane.

Stability Analysis: Results

Orbit Name	Largest Eigenvalue
Lagrange3	85.0
OrthQuasiEllipse4	18.3
Rosette4	1.7
Braid4	678.6
Trefoil4	41,306.0
FigureElght4	166.9
FoldedTriLoop4	74,895.0
PlateSaucer4	3,660.0
BorderCollie4	188.4
Trefoil5	191,300,000.0
FigureEight5	2,200.0

Orbit Name	Largest Eigenvalue
Lagrange2	1.0001
FigureEight3	1.0008
Ducati3	1.0003
Hill3_3	1.0104
Hill15	1.5906
DoubleDouble5	12.2980
DoubleDouble10	1.4040
DoubleDouble20	1.8900